

Pensieve header: Tristram-Levine signatures via braid closures. Continued
pensieve://Projects/Signatures/.

```
In[ ]:= << KnotTheory`
```

```
Loading KnotTheory` version of February 2, 2020, 10:53:45.2097.  
Read more at http://katlas.org/wiki/KnotTheory.
```

```
In[ ]:= K = Knot[3, 1];
```

```
K // PD
```

```
K // PD // Jones
```

```
K // PD // BR
```

```
K // PD // BR // PD
```

```
K // PD // BR // PD // Jones
```

```
Out[ ]:= PD[X[1, 4, 2, 5], X[3, 6, 4, 1], X[5, 2, 6, 3]]
```

```
Out[ ]:=  $-\frac{1}{\#1^4} + \frac{1}{\#1^3} + \frac{1}{\#1}$  &
```

```
Out[ ]:= BR[2, {-1, -1, -1}]
```

```
Out[ ]:= PD[X[6, 3, 1, 4], X[4, 1, 5, 2], X[2, 5, 3, 6]]
```

```
Out[ ]:=  $-\frac{1}{\#1^4} + \frac{1}{\#1^3} + \frac{1}{\#1}$  &
```

IN[i1, j1, i2, j2] measures the upper-half-plane intersection number of the arrows $i_1 \rightarrow j_1$ and $i_2 \rightarrow j_2$.

IN[1, 3, 2, 4] is +1.

```
In[ ]:= X[cond_] := If[TrueQ[cond], 1, 0];
```

```
IN[i1_, j1_, i2_, j2_] := -X[i1 > i2] + X[i1 > j2] + X[j1 > i2] - X[j1 > j2];
```

```
IN[1, 3, 2, 4]
```

```
Out[ ]:= 1
```

SC stands for "Simple Cycle".

```
In[ ]:=  $\beta$  = Knot[4, 1] // BR
```

```
{n =  $\beta$ [[1]], l = Length@ $\beta$ [[2]]}
```

```
Out[ ]:= BR[3, {-1, 2, -1, 2}]
```


```
Out[ ]:= {3, 4}
```

```
In[ ]:= SCs[ $\beta$ _BR] := Module[{n =  $\beta$ [[1]]},
```

```
Flatten@Table[SC@@@Subsets[Flatten@Position[Abs[ $\beta$ [[2]]], k], {2}], {k, n - 1}]]
```

```
In[ ]:= SCs[ $\beta$ ]
```

```
Out[ ]:= {SC[1, 3], SC[2, 4]}
```



$lk[\beta, sc1, sc2]$ computes $lk(sc1, sc2^+)$ in β .

```
In[ ]:= Clear[a1, a2, a3, a4];
{a1, a2, a3, a4} = {1, 0, -1, -1};
lk[ $\beta_{BR}$ , SC[i1_, j1_], SC[i2_, j2_]] := Module[
  {n =  $\beta$ [[1]], s1 = Abs@ $\beta$ [[2, i1]], s2 = Abs@ $\beta$ [[2, i2]]},
  Which[
    Abs[s2 - s1] > 1, 0,
    s2 - s1 == 1, a1 IN[i1, j1, j2, i2],
    s1 - s2 == 1, a2 IN[i1, j1, j2, i2],
    s1 == s2, a3 IN[i1 + 0.1, j1 + 0.1, i2, j2] +
      a4 (x[i1 == i2] * x[ $\beta$ [[2, i1]] > 0] - x[i1 == j2] * x[ $\beta$ [[2, i1]] > 0] -
        x[j1 == i2] * x[ $\beta$ [[2, j1]] > 0] + x[j1 == j2] * x[ $\beta$ [[2, j1]] > 0])
  ]
];
Table[lk[ $\beta$ , sc1, sc2], {sc1, SCs[ $\beta$ ]}, {sc2, SCs[ $\beta$ ]}] // MatrixForm
```

```
Out[ ]//MatrixForm=
( 1 -1 )
( 0 -1 )
```

SM for Seifert Matrix.

```

In[ ]:= SM[β_BR] := Module[{n = β[[1]], col, H1},
  H1 = Flatten@Table[
    col = Flatten@Position[Abs[β[[2]], k];
    SC[First@col, #] & /@Rest[col],
    {k, n - 1}];
  Table[lk[β, sc1, sc2], {sc1, H1}, {sc2, H1}]
];
SM[β] // MatrixForm

```

Out[]//MatrixForm=

$$\begin{pmatrix} 1 & -1 \\ 0 & -1 \end{pmatrix}$$

```

In[ ]:= SM[BR@Knot[6, 1]] // MatrixForm

```

Out[]//MatrixForm=

$$\begin{pmatrix} 1 & 0 & 0 & 0 \\ 1 & 0 & -1 & 0 \\ 0 & 0 & 1 & -1 \\ 0 & 0 & 0 & -1 \end{pmatrix}$$

```

In[ ]:= AlexNT[K_] := Module[{sm = SM[BR[K]]}, Det[t sm - Transpose@sm]]

```

```

In[ ]:= AlexNT[Knot[3, 1]]

```

Out[]:= $1 - t + t^2$

```

In[ ]:= Collect[AlexNT[Knot[6, 1]], t]

```

Out[]:= $-2t + 5t^2 - 2t^3$

```

In[ ]:= Alexander[Knot[6, 1]][t]

```

Out[]:= $5 - \frac{2}{t} - 2t$

```

In[ ]:= AlexNT[PD@BR[3, {1, 2, 1, 2}]]

```

Out[]:= $1 - t + t^2$

```

In[ ]:= Alexander[BR[3, {1, 2, 1, 2}]] [t]

```

Out[]:= $-1 + \frac{1}{t} + t$

```
In[ ]:= K = Knot[10, 100];
Alexander[K][t]
Collect[AlexNT[K], t]
Simplify[Alexander[K][t] / AlexNT[K]]
```

$$\text{Out[]} = 13 + \frac{1}{t^4} - \frac{4}{t^3} + \frac{9}{t^2} - \frac{12}{t} - 12t + 9t^2 - 4t^3 + t^4$$

$$\text{Out[]} = 1 - 4t + 9t^2 - 12t^3 + 13t^4 - 12t^5 + 9t^6 - 4t^7 + t^8$$

$$\text{Out[]} = \frac{1}{t^4}$$

```
In[ ]:= Union@Table[Simplify[Alexander[K][t] / AlexNT[K]], {K, AllKnots[]}]
```

Det: Argument {} at position 1 is not a non-empty square matrix.

$$\text{Out[]} = \left\{ \frac{1}{t^{21}}, \frac{1}{t^{14}}, \frac{1}{t^{13}}, \frac{1}{t^{12}}, \frac{1}{t^{10}}, \frac{1}{t^9}, \frac{1}{t^8}, \frac{1}{t^7}, \frac{1}{t^6}, \frac{1}{t^5}, \frac{1}{t^4}, \frac{1}{t^3}, \frac{1}{t^2}, \frac{1}{t}, \frac{1}{\text{Det}[\{\}]} \right\}$$

```
In[ ]:= MatrixSignature[M_?MatrixQ] := 2 Length@Select[Eigenvalues[M], # > 0 &] - Length[M]
```

```
In[ ]:= MatrixSignature /@ {{(1 0), (1 0)}, {(0 1), (0 -1)}, {(-1 0), (0 -1)}}
```

$$\text{Out[]} = \{2, 0, -2\}$$

```
In[ ]:= KSNT[K_] := Module[{M = SM[BR[K]]}, MatrixSignature[M + Transpose[M]]]
```

```
In[ ]:= KSNT /@ AllKnots[{3, 8}]
```

$$\text{Out[]} = \{2, 0, 4, 2, 0, 2, 0, 6, 2, -4, -2, 4, 2, 0, 0, 4, 0, 2, -4, 2, -2, 0, 0, -2, 2, 0, 0, 2, 4, 2, 0, 0, -6, 0, 2\}$$

```
In[ ]:= KnotSignature /@ AllKnots[{3, 8}]
```


$$\text{Out[]} = \{-2, 0, -4, -2, 0, -2, 0, -6, -2, 4, 2, -4, -2, 0, 0, -4, 0, -2, 4, -2, 2, 0, 0, 2, -2, 0, 0, -2, -4, -2, 0, 0, 6, 0, -2\}$$

```
In[ ]:= Union[(KSNT[#] + KnotSignature[#]) & /@ AllKnots[{3, 11}]]
```

$$\text{Out[]} = \{\emptyset\}$$

```
In[ ]:= TLS[K_, \omega_] := Module[{M = SM[BR[K]]}, MatrixSignature[(1 - \omega) M + (1 - Conjugate@\omega) Transpose[M]]]
```

```
In[ ]:= K = Knot[3, 1]; Plot[TLS[K, ei t], {t, 0, 2 π}]
```



```
In[ ]:= Table[Plot[TLS[K, ei t], {t, 0, 2 π}, PlotLabel -> K], {K, AllKnots[{3, 11}]}
```

